

Creating Simple Individual Reports for Study Participants

Elena-perkhounkova@uiowa.edu

April 24, 2014

The Goal

Patient ID=101

<i>Measurement</i>	<i>Visit 1</i>	<i>Visit 2</i>	<i>Visit 3</i>	<i>Change: Visit 1 to 2</i>	<i>Change: Visit 2 to 3</i>	<i>Change: Visit 1 to 3</i>
Weight (lbs)	143.6	135.2	142.6	-8.4	7.4	-1.0
BMI	23.0	21.6	22.8	-1.4	1.3	-0.2
Fat mass (lbs)	57.0	53.2	57.6	-3.9	4.4	0.6
Lean mass (lbs)	77.5	74.8	77.2	-2.8	2.4	-0.4
Percent fat	40.6	39.8	40.9	-0.8	1.1	0.3

SAS Data Set

	patient_id	weight_b	weight_m	weight_f	weightchfb	weightchfm	weightchmb
1	101	143.6	135.2	142.6	-1	7.4	-8.4
2	102	139.8
3	103	95.6	98.2	93.6	-2	-4.6	2.6
4	104	232.8	235.4	228.2	-4.6	-7.2	2.6
5	105	154
6	106	228
7	107	311	309.2	.	.	.	-1.8
8	108	194.4	191.8	185.2	-9.2	-6.6	-2.6
9	109	145.6	140	148.6	3	8.6	-5.6
10	110	174.4	175.4	171.4	-3	-4	1
11	111	270.8	264.4	270.2	-0.6	5.8	-6.4
12	112	287.2	289.6	281.8	-5.4	-7.8	2.4
13	113	194.4
14	114	185	185.4	184.6	-0.4	-0.8	0.4

A Macro to Create Lines for Each Measure

```

*macro creates a line for each measure;
*each line has 7 variables: measure's name,
  3 time points, 3 changes;
%macro measures(vn,v);
  measurement="&vn.          ";
  v1=round(&v._b,.1);
  v2=round(&v._m,.1);
  v3=round(&v._f,.1);
  v1t2=round(&v.chmb,.1);
  v2t3=round(&v.chfm,.1);
  v1t3=round(&v.chfb,.1);
  output;
%mend measures;

%measures(Weight (lbs),weight)

```

Using %Measures to Create a Data Set with a Line for Each Measure

```

*data step creates a data set with 5 lines per subject;
data indreports;
  set temp;
  %measures(Weight (lbs),weight)
  %measures(BMI,bmi)
  %measures(Fat mass (lbs),fatmass)
  %measures(Lean mass (lbs),leanmass)
  %measures(Percent fat,totpfat)
  label measurement="Measurement";
  label v1="Visit 1*";
  label v2="Visit 2*";
  label v3="Visit 3*";
  label v1t2="Change: Visit 1 to 2*";
  label v2t3="Change: Visit 2 to 3*";
  label v1t3="Change: Visit 1 to 3*";
  drop weight_b--fatmasschmb;
run;

```

SAS Data Set with a Line for Each Measure

	patient_id	measurement	v1	v2	v3	v1t2	v2t3	v1t3
1	101	Weight (lbs)	143.6	135.2	142.6	-8.4	7.4	-1
2	101	BMI	23	21.6	22.8	-1.4	1.3	-0.2
3	101	Fat mass (lbs)	57	53.2	57.6	-3.9	4.4	0.6
4	101	Lean mass (lbs)	77.5	74.8	77.2	-2.8	2.4	-0.4
5	101	Percent fat	40.6	39.8	40.9	-0.8	1.1	0.3
6	102	Weight (lbs)	139.8
7	102	BMI	22.7
8	102	Fat mass (lbs)	46.6
9	102	Lean mass (lbs)	85.5
10	102	Percent fat	33.7

PROC PRINT to Print Individual Reports

```

options number=no nodate;
*proc print prints individual reports;
proc print data=indreports noobs label split="*";
    var measurement v1 v2 v3 v1t2 v2t3 v1t3;
    by patient_id;
    pageby patient_id;
run;
/*Options in PROC PRINT:
NOOBS suppresses the printing of observation numbers
    at the beginning of the rows
LABEL uses variables' labels as column headings.
SPLIT='split-character' specifies the split character,
    which controls line breaks in column headings.*/

```

An Individual Report

Patient ID=101

<i>Measurement</i>	<i>Visit 1</i>	<i>Visit 2</i>	<i>Visit 3</i>	<i>Change: Visit 1 to 2</i>	<i>Change: Visit 2 to 3</i>	<i>Change: Visit 1 to 3</i>
Weight (lbs)	143.6	135.2	142.6	-8.4	7.4	-1.0
BMI	23.0	21.6	22.8	-1.4	1.3	-0.2
Fat mass (lbs)	57.0	53.2	57.6	-3.9	4.4	0.6
Lean mass (lbs)	77.5	74.8	77.2	-2.8	2.4	-0.4
Percent fat	40.6	39.8	40.9	-0.8	1.1	0.3